BEOSIN

Blockchain Security

Smart contract security audit report



Audit Number: 202107021148

Report Query Name: iLAVA&Airdrop

Smart Contract Name	Smart Contract Address	Smart Contract Address Link
iLAVAToken	0x924D79e9Ea369eb25491127daA9d42200f7c1aD0	https://bscscan.com/address/0x924D79e9Ea369eb25491127daA9d42200f7c1aD0#code
Airdrop	0xCAb6959eC8A55b57fD7D0671042364BEe1f7Ea6C	https://bscscan.com/address/0xCAb6959eC8A55b57fD7D0671042364BEe1f7Ea6C#code

Start Date: 2021.06.24

Completion Date: 2021.07.02

Overall Result: Pass

Audit Team: Beosin (Chengdu LianAn) Technology Co. Ltd.

Audit Categories and Results:

No.	Categories	Subitems	Results
1	Coding Conventions	Compiler Version Security	Pass
		Deprecated Items	Pass
		Redundant Code	Pass
		SafeMath Features	Pass
		require/assert Usage	Pass
		Gas Consumption	Pass
		Visibility Specifiers	Pass
		Fallback Usage	Pass
2	General Vulnerability	Integer Overflow/Underflow	Pass
		Reentrancy	Pass
		Pseudo-random Number Generator (PRNG)	Pass
		Transaction-Ordering Dependence	Pass
		DoS (Denial of Service)	Pass
		Access Control of Owner	Pass

		Low-level Function (call/delegatecall) Security	Pass
		Returned Value Security	Pass
		tx.origin Usage	Pass
		Replay Attack	Pass
		Overriding Variables	Pass
3	Business Security	Business Logics	Pass
		Business Implementations	Pass

Audit Results Explained:

Beosin (Chengdu LianAn) Technology has used several methods including Formal Verification, Static Analysis, Typical Case Testing and Manual Review to audit three major aspects of smart contracts project iLAVA&Airdrop, including Coding Standards, Security, and Business Logic. **The iLAVA&Airdrop project passed all audit items. The overall result is Pass. The smart contract is able to function properly.**

Audit Contents:

1. Coding Conventions

Check the code style that does not conform to Solidity code style.

1.1 Compiler Version Security

- Description: Check whether the code implementation of current contract contains the exposed solidity compiler bug.
- Result: Pass

1.2 Deprecated Items

- Description: Check whether the current contract has the deprecated items.
- Result: Pass

1.3 Redundant Code

- Description: Check whether the contract code has redundant codes.
- Result: Pass

1.4 SafeMath Features

- Description: Check whether the SafeMath has been used. Or prevents the integer overflow/underflow in mathematical operation.
- Result: Pass

1.5 require/assert Usage

- Description: Check the use reasonability of 'require' and 'assert' in the contract.
- Result: Pass

1.6 Gas Consumption

- Description: Check whether the gas consumption exceeds the block gas limitation.
- Result: Pass

1.7 Visibility Specifiers

- Description: Check whether the visibility conforms to design requirement.
- Result: Pass

1.8 Fallback Usage

- Description: Check whether the Fallback function has been used correctly in the current contract.
- Result: Pass

2. General Vulnerability

Check whether the general vulnerabilities exist in the contract.

2.1 Integer Overflow/Underflow

- Description: Check whether there is an integer overflow/underflow in the contract and the calculation result is abnormal.
- Result: Pass

2.2 Reentrancy

- Description: An issue when code can call back into your contract and change state, such as withdrawing BNB.
- Result: Pass

2.3 Pseudo-random Number Generator (PRNG)

- Description: Whether the results of random numbers can be predicted.
- Result: Pass

2.4 Transaction-Ordering Dependence

- Description: Whether the final state of the contract depends on the order of the transactions.
- Result: Pass

2.5 DoS (Denial of Service)

- Description: Whether exist DoS attack in the contract which is vulnerable because of unexpected reason.
- Result: Pass

2.6 Access Control of Owner

- Description: Whether the owner has excessive permissions, such as malicious issue, modifying the balance of others.
- Result: Pass

2.7 Low-level Function (call/delegatecall) Security

- Description: Check whether the usage of low-level functions like call/delegatecall have vulnerabilities.
- Result: Pass

2.8 Returned Value Security

- Description: Check whether the function checks the return value and responds to it accordingly.
- Result: Pass

2.9 tx.origin Usage

- Description: Check the use secure risk of 'tx.origin' in the contract.
- Result: Pass

2.10 Replay Attack

- Description: Check whether the implement possibility of Replay Attack exists in the contract.
- Result: Pass

2.11 Overriding Variables

- Description: Check whether the variables have been overridden and lead to wrong code execution.
- Result: Pass

3. Business Security

Check whether the business is secure.

3.1 Business analysis of Contract iLAVAToken

(1) Basic Token Information

Token name	iLAVA Membership Token
Token symbol	iLAVA
decimals	18
totalSupply	The initial supply is 0
Token type	BEP-20

Table 1 Basic Token Information

(2) BEP-20 Token Standard Functions

- Description: The token contract implements a token which conforms to the BEP-20 Standards. It should be noted that the user can directly call the *approve* function to set the approval value for the specified address, but in order to avoid multiple authorizations, it is recommended that the user resets the authorization value to 0 when calling this function to change the authorization value. Token transfer related functions are only available when the related status is true.

```
585     function totalSupply() public view returns (uint256 iLAVASupply) {
586         uint256 totalLAVA = IERC20(_LAVA_TOKEN_).balanceOf(address(this));
587         (,uint256 curDistribution) = getLatestAlpha();
588         uint256 actualLAVA = totalLAVA.sub(_TOTAL_BLOCK_REWARD_.sub(curDistribution.add(_TOTAL_BLOCK_DISTRIBUTION_)));
589         iLAVASupply = actualLAVA / _LAVA_RATIO_;
590     }
591
592     function balanceOf(address account) public view returns (uint256 iLAVAAmount) {
593         iLAVAAmount = lavaBalanceOf(account) / _LAVA_RATIO_;
594     }
595
596     function transfer(address to, uint256 iLAVAAmount) public returns (bool) {
597         _updateAlpha();
598         _transfer(msg.sender, to, iLAVAAmount);
599         return true;
600     }
601
602     function approve(address spender, uint256 iLAVAAmount) canTransfer public returns (bool) {
603         _ALLOWED_[msg.sender][spender] = iLAVAAmount;
604         emit Approval(msg.sender, spender, iLAVAAmount);
605         return true;
606     }
607
608     function transferFrom(
609         address from,
610         address to,
611         uint256 iLAVAAmount
612     ) public returns (bool) {
613         require(iLAVAAmount <= _ALLOWED_[from][msg.sender], "ALLOWANCE_NOT_ENOUGH");
614         _updateAlpha();
615         _transfer(from, to, iLAVAAmount);
616         _ALLOWED_[from][msg.sender] = _ALLOWED_[from][msg.sender].sub(iLAVAAmount);
617         return true;
618     }
619
620     function allowance(address owner, address spender) public view returns (uint256) {
621         return _ALLOWED_[owner][spender];
622     }
```

Figure 1 source code of BEP-20 functions

Note: The total supply of tokens is calculated by the formula, and the relevant parameters are modified and not checked before updating, which may result in a subtractive overflow in the calculation, e.g. _TOTAL_BLOCK_REWARD_ may be smaller than the curDistribution that increases over time(as figure 2, 3 below). and, the owner of the contract can set _LAVA_RATIO_ causing the user's balance display to change.

```
function totalSupply() public view returns (uint256 iLAVASupply) {
    uint256 totalLAVA = IERC20(_LAVA_TOKEN_).balanceOf(address(this));
    (,uint256 curDistribution) = getLatestAlpha();
    uint256 actualLAVA = totalLAVA.sub(_TOTAL_BLOCK_REWARD_.sub(curDistribution.add(_TOTAL_BLOCK_DISTRIBUTION_)));
    iLAVASupply = actualLAVA / _LAVA_RATIO_;
}
```

Figure 2 source code of *totalSupply*

Figure 3 error about *totalSupply*

```
function _transfer(
    address from,
    address to,
    uint256 iLAVAAmount
) internal canTransfer balanceEnough(from, iLAVAAmount) {
    require(from != address(0), "transfer from the zero address");
    require(to != address(0), "transfer to the zero address");
    require(from != to, "transfer from same with to");

    uint256 stakingPower = DecimalMath.divFloor(iLAVAAmount * _LAVA_RATIO_, alpha);

    UserInfo storage fromUser = userInfo[from];
    UserInfo storage toUser = userInfo[to];

    _redeem(fromUser, stakingPower);
    _mint(toUser, stakingPower);

    emit Transfer(from, to, iLAVAAmount);
}
```

Figure 4 source code of *_transfer*

- Related functions: *name, symbol, decimals, totalSupply, balanceOf, allowance, transfer, transferFrom, approve*
- Safety recommendation: It is suggested to modify the calculation formula.
- Repair result: Fixed. The *getLatestAlpha* function has been modified so that when _TOTAL_BLOCK_REWARD is not greater than _TOTAL_BLOCK_DISTRIBUTION_, the value of alpha will no longer change and the total amount of tokens and the user's balance will not change and no error will occur.

```
620     function getLatestAlpha() public view returns (uint256 newAlpha, uint256 curDistribution) {
621         if (_LAST_REWARD_BLOCK_ == 0) {
622             curDistribution = 0;
623         } else {
624             // curDistribution = _LAVA_PER_BLOCK_ * (block.number - _LAST_REWARD_BLOCK_);
625             if(_TOTAL_BLOCK_REWARD_ <= _TOTAL_BLOCK_DISTRIBUTION_){
626                 curDistribution = 0;
627             }
628             else{
629                 uint256 _curDistribution = _LAVA_PER_BLOCK_ * (block.number - _LAST_REWARD_BLOCK_);
630                 uint256 diff = _TOTAL_BLOCK_REWARD_.sub(_TOTAL_BLOCK_DISTRIBUTION_);
631                 curDistribution = diff < _curDistribution ? diff : _curDistribution;
632             }
633         }
634         if (_TOTAL_STAKING_POWER_ > 0) {
635             newAlpha = uint256(alpha).add(DecimalMath.divFloor(curDistribution, _TOTAL_STAKING_POWER_));
636         } else {
637             newAlpha = alpha;
638         }
639     }
```

Figure 5 source code of *getLatestAlpha* (new)

- Result: Pass

(3) mint function

- Description: The contract implements the *mint* function for user participation in staking mining (requires pre-authorization of this contract). The first call to this function will carry out the registration of the user address, the superior address cannot be 0 and the caller itself, and the staking amount needs to be greater than 0; the internal function *_updateAlpha* will be called before the collateral to update the relevant data, and *_mint* will be called after the collateral to update the relevant data of superior address. If the airdropController address is not 0, the *deposit* function in the airdrop contract will be executed to update the airdrop reward related parameters.

```
463     function mint(uint256 lavaAmount, address superiorAddress) public {
464         require(
465             superiorAddress != address(0) && superiorAddress != msg.sender,
466             "iLAVAToken: Superior INVALID"
467         );
468         require(lavaAmount > 0, "iLAVAToken: must mint greater than 0");
469 
470         UserInfo storage user = userInfo[msg.sender];
471 
472         if (user.superior == address(0)) {
473             require(
474                 superiorAddress == _LAVA_TEAM_ || userInfo[superiorAddress].superior != address(0),
475                 "iLAVAToken: INVALID_SUPERIOR_ADDRESS"
476             );
477             user.superior = superiorAddress;
478         }
479 
480         _updateAlpha();
481 
482         IERC20(_LAVA_TOKEN_).transferFrom(msg.sender, address(this), lavaAmount);
483 
484         uint256 newStakingPower = DecimalMath.divFloor(lavaAmount, alpha);
485 
486         _mint(user, newStakingPower);
487 
488         user.originAmount = user.originAmount.add(lavaAmount);
489 
490         if(!isUser[msg.sender]){
491             isUser[msg.sender] = true;
492             totalUsers = totalUsers.add(1);
493         }
494 
495         if(address(airdropController) != address(0)){
496             airdropController.deposit(msg.sender, newStakingPower);
497         }
498 
499 
500         emit MintILAVA(msg.sender, superiorAddress, lavaAmount);
501     }
```

Figure 6 source code of *mint*

- Related functions: *mint, transferFrom, deposit*
- Result: Pass

(4) Ownership

- Description: The contract implements *transferOwnership* and *claimOwnership* functions to manage the contract's ownership. *transferOwnership* is used to set the newOwner address and can only be called by the current owner of the contract; The *claimOwnership* function can be called only by the current newOwner to receive the ownership and reset the newOwner address to 0.

```
function transferOwnership(address newOwner) public onlyOwner {
    emit OwnershipTransferPrepared(_OWNER_, newOwner);
    _NEW_OWNER_ = newOwner;
}

function claimOwnership() public {
    require(msg.sender == _NEW_OWNER_, "INVALID_CLAIM");
    emit OwnershipTransferred(_OWNER_, _NEW_OWNER_);
    _OWNER_ = _NEW_OWNER_;
    _NEW_OWNER_ = address(0);
}
```

Figure 7 source code of *transferOwnership* and *claimOwnership*

- Related functions: *transferOwnership, claimOwnership*
- Result: Pass

(5) Initialize owner

- Description: The contract implements the *initOwner* function to initialize the owner after the contract is deployed and can only be called once. It is recommended to call the contract immediately after it is deployed.

```
212     function initOwner(address newOwner) public notInitialized {
213         _INITIALIZED_ = true;
214         _OWNER_ = newOwner;
215     }
```

Figure 8 source code of *initOwner*

- Related functions: *initOwner*
- Result: Pass

(6) Donate

- Description: The contract implements the *donate* function for users to donate tokens to the contract, which will update the value of alpha.

```
568     function donate(uint256 lavaAmount) public {
569         IERC20(_LAVA_TOKEN_).transferFrom(msg.sender, address(this), lavaAmount);
570 
571         alpha = uint112(
572             uint256(alpha).add(DecimalMath.divFloor(lavaAmount, _TOTAL_STAKING_POWER_))
573         );
574         emit DonateLAVA(msg.sender, lavaAmount);
575     }
```

Figure 9 source code of *donate*

- Related functions: *donate*
- Result: Pass

(7) Redeem

- Description: The contract implements the *redeem* function for the user to withdraw the pledged Lava tokens. Before the withdrawal, the internal function *_updateAlpha* is called to update the relevant data, determine whether the user is withdrawing all, call the internal function *_redeem* to update the information about the superior address. Then calculate the actual withdrawal amount, whether destruction and transaction fees are incurred, and make the relevant transfer. If the user withdraws all, the user's identity will be cancelled. If the airdropController address is not 0, the *withdraw* function in the airdrop contract will be executed to update the airdrop reward related parameters.

```
503     function redeem(uint256 ilavaAmount, bool all) public balanceEnough(msg.sender, ilavaAmount) {
504
505         _updateAlpha();
506         UserInfo storage user = userInfo[msg.sender];
507
508         uint256 lavaAmount;
509         uint256 stakingPower;
510
511         if (all) {
512             stakingPower = uint256(user.stakingPower).sub(DecimalMath.divFloor(user.credit, alpha));
513             lavaAmount = DecimalMath.mulFloor(stakingPower, alpha);
514         } else {
515             lavaAmount = ilavaAmount.mul(_LAVA_RATIO_);
516             stakingPower = DecimalMath.divFloor(lavaAmount, alpha);
517         }
518
519         _redeem(user, stakingPower);
520
521         (uint256 lavaReceive, uint256 burnLAVAAmount, uint256 withdrawFeeLAVAAmount) = getWithdrawResul
522
523         IERC20(_LAVA_TOKEN_).transfer(msg.sender, lavaReceive);
524
525         if (burnLAVAAmount > 0) {
526             IERC20(_LAVA_TOKEN_).transfer(_LAVA_BURN_ADDRESS_, burnLAVAAmount);
527         }
528
529         if (withdrawFeeLAVAAmount > 0) {
530             alpha = uint112(
531                 uint256(alpha).add(
532                     DecimalMath.divFloor(withdrawFeeLAVAAmount, _TOTAL_STAKING_POWER_)
533                 )
534             );
535         }
536
537         if (withdrawFeeLAVAAmount > 0) {
538             totalWithdrawFee = totalWithdrawFee.add(withdrawFeeLAVAAmount);
539         }
540
541         if(burnLAVAAmount > 0){
542             totalBurnLAVA = totalBurnLAVA.add(burnLAVAAmount);
543         }
544
545         if(user.originAmount <= lavaAmount){
546             user.originAmount = 0;
547         }
548         else{
549             user.originAmount = user.originAmount.sub(lavaAmount);
550         }
551
552         if(all){
553             if(isUser[msg.sender]){
554                 isUser[msg.sender] = false;
555                 if(totalUsers > 0){
556                     totalUsers = totalUsers.sub(1);
557                 }
558             }
559         }
560
561         if(address(airdropController) != address(0)){
562             airdropController.withdraw(msg.sender, stakingPower);
563         }
564
565         emit RedeemILAVA(msg.sender, lavaReceive, burnLAVAAmount, withdrawFeeLAVAAmount);
566     }
```

Figure 10 source code of *redeem*

- Related functions: *redeem, withdraw*
- Result: Pass

(8) Pre-deposit

- Description: The contract implements a *preDepositedBlockReward* for users to send Lava tokens as reward, this part of Lava tokens will not enter into iLAVA related calculations and cannot be withdrawn.

```
577     function preDepositedBlockReward(uint256 lavaAmount) public {
578         IERC20(_LAVA_TOKEN_).transferFrom(msg.sender, address(this), lavaAmount);
579         _TOTAL_BLOCK_REWARD_ = _TOTAL_BLOCK_REWARD_.add(lavaAmount);
580         emit PreDeposit(lavaAmount);
581     }
```

Figure 11 source code of *preDepositedBlockReward*

- Related functions: *preDepositedBlockReward*
- Result: Pass

(9) Contract parameter setting functions

- Description: The contract implements the following functions that only the contract owner can call: The *setAirdropController* function is used to set the address of the airdropController contract; *setCantransfer* to set whether iLAVA transfers are allowed; *changePerReward* to change _LAVA_PER_BLOCK_; *updateLAVAFeeBurnRatio* to change the rate of the destruction fee. *updateLAVAFeeBurnAddress* for setting the address to receive tokens when they are destroyed; *updateGovernance* for setting _DOOD_GOV_; *updateSuperiorRatio* for setting the rate of reward for superior addresses; *updateFeeRatio* for setting the rate of transaction fees; *emergencyWithdraw* is used to withdraw all Lava tokens from the contract to the owner's address. Note: The owner can extract all the Lava tokens in the contract by calling the *emergencyWithdraw* function, which may affect subsequent users calling redeem to redeem their Lava tokens. And when modifying the relevant parameters without judging whether the parameters are appropriate, the modification may lead to errors in the relevant calculation.

```
420     function setAirdropController(address _controller) public onlyOwner {
421         airdropController = IAirdrop(_controller);
422     }
423
424     function setCantransfer(bool allowed) public onlyOwner {
425         _CAN_TRANSFER_ = allowed;
426         emit SetCantransfer(allowed);
427     }
428
429     function changePerReward(uint256 lavaPerBlock) public onlyOwner {
430         _updateAlpha();
431         _LAVA_PER_BLOCK_ = lavaPerBlock;
432         emit ChangePerReward(lavaPerBlock);
433     }
434
435     function updateLAVAFeeBurnRatio(uint256 lavaFeeBurnRatio) public onlyOwner {
436         _LAVA_FEE_BURN_RATIO_ = lavaFeeBurnRatio;
437         emit UpdateLAVAFeeBurnRatio(_LAVA_FEE_BURN_RATIO_);
438     }
439
440     function updateLAVAFeeBurnAddress(address addr) public onlyOwner{
441         _LAVA_BURN_ADDRESS_ = addr;
442     }
443
444     function updateGovernance(address governance) public onlyOwner {
445         _DOOD_GOV_ = governance;
446     }
447
448     function updateSuperiorRatio(uint256 superiorRatio) public onlyOwner {
449         _SUPERIOR_RATIO_ = superiorRatio;
450     }
451
452     function updateFeeRatio(uint256 feeRatio) public onlyOwner {
453         _FEE_RATIO = feeRatio;
454     }
455
456     function emergencyWithdraw() public onlyOwner {
457         uint256 lavaBalance = IERC20(_LAVA_TOKEN_).balanceOf(address(this));
458         IERC20(_LAVA_TOKEN_).transfer(_OWNER_, lavaBalance);
459     }
```

Figure 12 source code of Ownable functions

- Related functions: *setAirdropController, setCantransfer, changePerReward, updateLAVAFeeBurnRatio, updateLAVAFeeBurnAddress, updateGovernance, updateSuperiorRatio, updateFeeRatio, emergencyWithdraw*
- Safety recommendation: The *emergencyWithdraw* function has excessive owner privileges and can extract Lava tokens pledged by the user, so it is recommended to remove it. The *updateFeeRatio* has excessive owner privileges and can set the fee rate arbitrarily and the fee receiving address can be set freely by the owner. It is recommended to add a limit to the fee rate to prevent malicious tokens from being sent to a specific address after the private key is lost.
- Repair result: Deleted and increased maximum transaction fee rate (20%).

```
417     // ============ Ownable Functions ============`
418
419     function setAirdropController(address _controller) public onlyOwner {
420         airdropController = IAirdrop(_controller);
421     }
422
423     function setCantransfer(bool allowed) public onlyOwner {
424         _CAN_TRANSFER_ = allowed;
425         emit SetCantransfer(allowed);
426     }
427
428     function changePerReward(uint256 lavaPerBlock) public onlyOwner {
429         _updateAlpha();
430         _LAVA_PER_BLOCK_ = lavaPerBlock;
431         emit ChangePerReward(lavaPerBlock);
432     }
433
434     function updateLAVAFeeBurnRatio(uint256 lavaFeeBurnRatio) public onlyOwner {
435         _LAVA_FEE_BURN_RATIO_ = lavaFeeBurnRatio;
436         emit UpdateLAVAFeeBurnRatio(_LAVA_FEE_BURN_RATIO_);
437     }
438
439     function updateLAVAFeeBurnAddress(address addr) public onlyOwner{
440         _LAVA_BURN_ADDRESS_ = addr;
441     }
442
443     function updateGovernance(address governance) public onlyOwner {
444         _DOOD_GOV_ = governance;
445     }
446
447     function updateSuperiorRatio(uint256 superiorRatio) public onlyOwner {
448         _SUPERIOR_RATIO_ = superiorRatio;
449     }
450
451     function updateFeeRatio(uint256 feeRatio) public onlyOwner {
452         require(feeRatio <= _MAX_FEE_RATIO, "_FEE_RATIO exceeded");
453         _FEE_RATIO = feeRatio;
454     }
455
```

Figure 13 source code of owner functions(new)

- Result: Pass

(10) Related parameter query function

- Description: The contract implements *getLatestAlpha* function to query the latest alpha value; *availableBalanceOf* function to query the available balance of the specified address; *lavaBalanceOf* function to calculate the number of Lava tokens pledged to the contract from the specified address; *getWithdrawResult* function to calculate the actual withdrawal amount based on the input amount; *getLAVAWithdrawFeeRatio* function to query the fee ratio of the Lava tokens withdrawn from the specified address; *getSuperior* function to query the superior address; *getWithdrawResult* function is used to calculate the actual number of tokens withdrawn based on the amount entered; *getLAVAWithdrawFeeRatio* function is used to query the fee rate for withdrawing Lava tokens; *getSuperior* function is used to query the superior address of the specified address; The *getUserStakingPower* function is used to query the collateral power of the specified address.



```
function getLatestAlpha() public view returns (uint256 newAlpha, uint256 curDistribution) {
    if (_LAST_REWARD_BLOCK_ == 0) {
        curDistribution = 0;
    } else {
        // curDistribution = _LAVA_PER_BLOCK_ * (block.number - _LAST_REWARD_BLOCK_);
        if(_TOTAL_BLOCK_REWARD_ <= _TOTAL_BLOCK_DISTRIBUTION_){
            curDistribution = 0;
        }
        else{
            uint256 _curDistribution = _LAVA_PER_BLOCK_ * (block.number - _LAST_REWARD_BLOCK_);
            uint256 diff = _TOTAL_BLOCK_REWARD_.sub(_TOTAL_BLOCK_DISTRIBUTION_);
            curDistribution = diff < _curDistribution ? diff : _curDistribution;
        }
    }
    if (_TOTAL_STAKING_POWER_ > 0) {
        newAlpha = uint256(alpha).add(DecimalMath.divFloor(curDistribution, _TOTAL_STAKING_POWER_));
    } else {
        newAlpha = alpha;
    }
}

function availableBalanceOf(address account) public view returns (uint256 iLAVAAmount) {
    if (_DOOD_GOV_ == address(0)) {
        iLAVAAmount = balanceOf(account);
    } else {
        uint256 lockediLAVAAmount = IGovernance(_DOOD_GOV_).getLockediLAVA(account);
        iLAVAAmount = balanceOf(account).sub(lockediLAVAAmount);
    }
}

function lavaBalanceOf(address account) public view returns (uint256 lavaAmount) {
    UserInfo memory user = userInfo[account];
    (uint256 newAlpha,) = getLatestAlpha();
    uint256 nominalLAVA =  DecimalMath.mulFloor(uint256(user.stakingPower), newAlpha);
    if(nominalLAVA > user.credit) {
        lavaAmount = nominalLAVA - user.credit;
    }else {
        lavaAmount = 0;
    }
}

function getWithdrawResult(uint256 lavaAmount)
    public
    view
    returns (
        uint256 lavaReceive,
        uint256 burnLAVAAmount,
        uint256 withdrawFeeLAVAAmount
    )
{
    uint256 feeRatio = _FEE_RATIO;

    withdrawFeeLAVAAmount = DecimalMath.mulFloor(lavaAmount, feeRatio);
    lavaReceive = lavaAmount.sub(withdrawFeeLAVAAmount);

    burnLAVAAmount = DecimalMath.mulFloor(withdrawFeeLAVAAmount, _LAVA_FEE_BURN_RATIO_);
    withdrawFeeLAVAAmount = withdrawFeeLAVAAmount.sub(burnLAVAAmount);
}

function getLAVAWithdrawFeeRatio() public view returns (uint256) {
    return _FEE_RATIO;
}

function getSuperior(address account) public view returns (address superior) {
    return userInfo[account].superior;
}

function getUserStakingPower(address account) public view returns (uint256){
    return userInfo[account].stakingPower;
}
```

Figure 14 source code of query functions

- Related functions: *getLatestAlpha, availableBalanceOf, lavaBalanceOf, getWithdrawResult, getLAVAWithdrawFeeRatio, getSuperior, getUserStakingPower*
- Result: Pass

3.2 Business analysis of Contract Token Airdrop

iLAVA's collateral arithmetic varies according to its holdings. iLAVA token species only *mint* and *redeem* functions update the user airdrop reward calculations in the Airdrop contract. If the iLAVA token is opened for transfer, the receiving address can update the data related to the airdrop reward through functions such as *syncIlava* to get the airdrop reward; however, the data related to the airdrop reward in Airdrop for the transferring address will not be updated and can continue to maintain the same yield as before the transfer.

(i.e. the iLAVA token holdings decrease while the reward remains unchanged) The project owner declares that iLAVA transfers will not be activated and that if they are, the relevant airdrop contract will be voided.

(1) add function

- Description: The contract implements the *add* function for the contract's owner to add new airdrop tokens and set airdrop reward related parameters. Note: Adding duplicate airdrop tokens will cause the reward to be calculated incorrectly, so administrators should be careful to prevent duplicate additions.

```
1207    function add(
1208        IERC20 _airdropToken,
1209        uint256 _airdropPerBlock,
1210        uint256 _startBlock,
1211        uint256 _finishBlock
1212        ) public onlyOwner {
1213        uint256 lastRewardBlock = block.number > _startBlock ? block.number : _startBlock;
1214        poolInfo.push(PoolInfo({
1215            airdropToken: _airdropToken,
1216            lastRewardBlock: lastRewardBlock,
1217            accSushiPerShare: 0,
1218            startBlock: _startBlock,
1219            finishBlock: _finishBlock,
1220            airdropPerBlock: _airdropPerBlock,
1221            lavaSupply: 0
1222        }));
1223    }
```

Figure 15 source code of *add* function

- Safety recommendation: It is recommended to add a finishBlock greater than the current time to prevent the reward from being calculated incorrectly.
- Repair result: Fixed

```
1207    function add(
1208        IERC20 _airdropToken,
1209        uint256 _airdropPerBlock,
1210        uint256 _startBlock,
1211        uint256 _finishBlock
1212        ) public onlyOwner {
1213        require(_finishBlock > block.number, "had finished");
1214        uint256 lastRewardBlock = block.number > _startBlock ? block.number : _startBlock;
1215        poolInfo.push(PoolInfo({
1216            airdropToken: _airdropToken,
1217            lastRewardBlock: lastRewardBlock,
1218            accSushiPerShare: 0,
1219            startBlock: _startBlock,
1220            finishBlock: _finishBlock,
1221            airdropPerBlock: _airdropPerBlock,
1222            lavaSupply: 0
1223        }));
1224    }
```

Figure 16 source code of *add* function(new)

- Related functions: *add*
- Result: Pass

(2) set function

- Description: The contract implements the *set* function for the owner of the contract to modify the parameters related to the airdrop token rewards for the specified id, optionally executing the *updatePool* function to update the rewards related data before the modification.

```
1225    function set(
1226        uint256 _pid,
1227        uint256 _airdropPerBlock,
1228        uint256 _startBlock,
1229        uint256 _finishBlock,
1230        bool _withUpdate
1231        ) public onlyOwner{
1232            if(_withUpdate){
1233                updatePool(_pid);
1234            }
1235            PoolInfo storage pool = poolInfo[_pid];
1236            pool.startBlock = _startBlock;
1237            pool.finishBlock = _finishBlock;
1238            pool.airdropPerBlock = _airdropPerBlock;
1239        }
```

Figure 17 source code of *set* function

- Safety recommendation: It is recommended to add a finishBlock greater than the current time to prevent the reward from being calculated incorrectly.
- Repair result: Fixed

```
1226    function set(
1227        uint256 _pid,
1228        uint256 _airdropPerBlock,
1229        uint256 _startBlock,
1230        uint256 _finishBlock,
1231        bool _withUpdate
1232        ) public onlyOwner{
1233            require(_finishBlock > block.number, "had finished");
1234            if(_withUpdate){
1235                updatePool(_pid);
1236            }
1237            PoolInfo storage pool = poolInfo[_pid];
1238            pool.startBlock = _startBlock;
1239            pool.finishBlock = _finishBlock;
1240            pool.airdropPerBlock = _airdropPerBlock;
1241        }
```

Figure 18 source code of *set* function(new)

- Related functions: *set, updatePool*
- Result: Pass

(3) updatePool function

- Description: The contract implements *updatePool* function to update the data related to the airdrop token rewards for the specified id.

```
1263    function updatePool(uint256 _pid) public{
1264
1265        PoolInfo storage pool = poolInfo[_pid];
1266
1267        uint256 currentBlockNumber = block.number > pool.finishBlock ? pool.finishBlock : block.number;
1268
1269        if (currentBlockNumber <= pool.lastRewardBlock) {
1270            return;
1271        }
1272        if(currentBlockNumber < pool.startBlock){
1273            return;
1274        }
1275        if (pool.lavaSupply == 0) {
1276            pool.lastRewardBlock = currentBlockNumber;
1277            return;
1278        }
1279        uint256 multiplier = getMultiplier(pool.lastRewardBlock, currentBlockNumber);
1280        uint256 airdropReward = multiplier.mul(pool.airdropPerBlock);
1281        pool.accSushiPerShare = pool.accSushiPerShare.add(airdropReward.mul(1e12).div(pool.lavaSupply));
1282        pool.lastRewardBlock = currentBlockNumber;
1283    }
```

Figure 19 source code of *updatePool* function

- Related functions: *updatePool, getMultiplier*
- Result: Pass

(4) deposit function

- Description: The contract implements the *deposit* function to update all the user's drop reward related data (increasing the user's calculation), by calling the internal function *_deposit,* and the updatePool is executed to update the airdrop token data before increasing. If the user's calculated amount is not 0, the previous airdrop rewards are calculated and sent. Only iLAVA token contract addresses can be called.

```
1286    function deposit(address account, uint256 _amount) onlyIlava public {
1287        for (uint256 _pid = 0; _pid < poolInfo.length; _pid++) {
1288            _deposit(account, _pid, _amount);
1289        }
1290    }
1291
1292
1293    function _deposit(address account, uint256 _pid, uint256 _amount) internal {
1294        PoolInfo storage pool = poolInfo[_pid];
1295        UserInfo storage user = userInfo[_pid][account];
1296        updatePool(_pid);
1297        uint256 pending = 0;
1298        if (user.amount > 0) {
1299            pending = user.amount.mul(pool.accSushiPerShare).div(1e12).sub(user.rewardDebt);
1300        }
1301        pool.lavaSupply = pool.lavaSupply.add(_amount);
1302        user.amount = user.amount.add(_amount);
1303        user.rewardDebt = user.amount.mul(pool.accSushiPerShare).div(1e12);
1304        if(pending > 0){
1305            safeAirdropTransfer(pool.airdropToken, account, pending);
1306        }
1307        emit Deposit(account, _pid, _amount);
1308    }
```

Figure 20 source code of *deposit* function

- Related functions: *deposit, updatePool, safeAirdropTransfer*
- Result: Pass

(5) withdraw function

- Description: The contract implements the *withdraw* function to update all the user's airdrop reward data (reducing the amount of calculations for the user), before reducing the *updatePool* to update the

airdrop token data. If the user's calculated amount is not 0, the previous airdrop rewards are calculated and sent. Only iLAVA token contract addresses can be called.

```
1311    function withdraw(address account, uint256 amount) onlyIlava public {
1312        for (uint256 _pid = 0; _pid < poolInfo.length; _pid++){
1313            uint256 _amount = amount;
1314            PoolInfo storage pool = poolInfo[_pid];
1315            UserInfo storage user = userInfo[_pid][account];
1316            updatePool(_pid);
1317            uint256 pending = user.amount.mul(pool.accSushiPerShare).div(1e12).sub(user.rewardDebt);
1318            if(user.amount < _amount){
1319                _amount = user.amount;
1320            }
1321            user.amount = user.amount.sub(_amount);
1322            user.rewardDebt = user.amount.mul(pool.accSushiPerShare).div(1e12);
1323            if(pool.lavaSupply < _amount){
1324                _amount = pool.lavaSupply;
1325            }
1326            pool.lavaSupply = pool.lavaSupply.sub(_amount);
1327            safeAirdropTransfer(pool.airdropToken, account, pending);
1328            emit Withdraw(account, _pid, _amount);
1329        }
1330    }
```

Figure 21 source code of *withdraw* function

- Related functions: *withdraw, updatePool, safeAirdropTransfer*
- Result: Pass

(6) sync functions

- Description: The contract implements the *syncIlava* function for the user to update the reward-related data for their specified airdrop tokens, calling the internal function *_deposit* to update when the user's iLAVA collateral arithmetic exceeds the amount of calculations for the specified airdrop tokens. *synctIlavaAll* function for the user to update the reward-related data for all their airdrop tokens, traversing all airdrop tokens and updating only iLAVA collateral arithmetic exceeds the computed amount of the corresponding airdrop token.

```
1332    function syncIlava(uint256 _pid) public {
1333        UserInfo storage user = userInfo[_pid][msg.sender];
1334        uint256 stakingPower = ilava.getUserStakingPower(msg.sender);
1335        if(stakingPower > user.amount){
1336            _deposit(msg.sender, _pid, stakingPower.sub(user.amount));
1337        }
1338    }
1339
1340    function synctIlavaAll() public{
1341        uint256 stakingPower = ilava.getUserStakingPower(msg.sender);
1342        for (uint256 _pid = 0; _pid < poolInfo.length; _pid++) {
1343            UserInfo storage user = userInfo[_pid][msg.sender];
1344            if(stakingPower > user.amount){
1345                _deposit(msg.sender, _pid, stakingPower.sub(user.amount));
1346            }
1347        }
1348    }
```

Figure 22 source code of sync functions

- Related functions: *syncIlava, synctIlavaAll, getUserStakingPower*
- Result: Pass

(7) harvest functions

- Description: The contract implements the *harvest* function for the user to receive the airdrop reward for the specified airdrop token, implemented by calling the internal function *_deposit*. The *harvestAll* function is used for the user to receive the airdrop reward for all airdrop tokens.

```
1351    function harvest(uint256 _pid) public{
1352        _deposit(msg.sender, _pid, 0);
1353    }
1354
1355    function harvestAll() public{
1356        for (uint256 _pid = 0; _pid < poolInfo.length; _pid++) {
1357            harvest(_pid);
1358        }
1359    }
```

Figure 23 source code of harvest functions

- Related functions: *harvest, harvestAll*
- Result: Pass

4. Conclusion

Beosin(ChengduLianAn) conducted a detailed audit on the design and code implementation of the smart contracts project iLAVA&Airdrop. The problems found by the audit team during the audit process have been notified to the project party and reached an agreement on the repair results, the overall audit result of the iLAVA&Airdrop project's smart contract is **Pass**.

Official Website

https://lianantech.com

E-mail

vaas@lianantech.com

Twitter

https://twitter.com/Beosin_com